Asymptotics of sequence A002513

(*Václav Kotěšovec, published Aug 24 2019*)

The sequence A002513 in OEIS (see [1]) is originally defined as an expansion of product

$$\prod_{k=1}^{\infty} \frac{1}{(1-x^{2k})^2 * (1-x^{2k-1})}$$

in powers of x.

Main result:

$$A002513(n) \sim \frac{e^{\pi\sqrt{n}}}{8\,n^{5/4}} * \left(1 - \frac{\frac{\pi}{16} + \frac{15}{8\,\pi}}{\sqrt{n}}\right)$$

Proof:

The generating function is also

$$\prod_{k=1}^{\infty} \frac{1}{(1-x^{k}) * (1-x^{2k})}$$

In the notation from [2] we have a formula for the main asymptotic term:

```
convsubexpfun[partminus[1, 1], partminus[2, 2]]
```

$$\frac{e^{\sqrt{n}\,\pi}}{8\,n^{5/4}}$$

For a minor asymptotic of the convolution of two sub-exponential functions I obtained in 2017 (using a several steps of "Series" in Mathematica) the general formula:

Let

$$f1(n) \sim \frac{e^{r1\sqrt{n}}}{n^{b1}} * \left(1 + \frac{c1}{\sqrt{n}}\right)$$

and

$$f2(n) \sim \frac{e^{r2\sqrt{n}}}{n^{b2}} * \left(1 + \frac{c2}{\sqrt{n}}\right)$$

where $r1 > 0, b1, c1$ and $r2 > 0, b2, c2$ are constants, then convolution of $f1(n)$ and $f2(n)$ is asymptotic to

```
convsubexpfun[Exp[r1*Sqrt[n]]/n^b1, Exp[r2*Sqrt[n]]/n^b2] * (1 + minorsqrt[r1,b1,c1,r2,b2,c2]/Sqrt[n])
```

where `convsubexpfun` see [2] and

```
minorsqrt[r1_, b1_, c1_, r2_, b2_, c2_] :=
  Simplify[(c1/r1 + c2/r2) * Sqrt[(r1^2 + r2^2)] +
    (2 * (b1 + b2 + (b1 - b2)^2) + (2 b1 - 1) * b1 * (r2^2 - r1^2) / r1^2 +
      (2 b2 - 1) * b2 * (r1^2 - r2^2) / r2^2 - 15 / 8) / Sqrt[(r1^2 + r2^2)]];
```

$$\prod_{k=1}^{\infty} \frac{1}{(1-x^k)}$$

is the generating function for a partitions (see A000041).
From Hardy-Ramanujan-Rademacher formula follows

```
(* A000041(n) minor asymptotic terms *)
```

$$\frac{1}{4\sqrt{3}\,n} e^{\sqrt{\frac{2}{3}}\sqrt{n}\,\pi} \left(1 - \frac{\frac{\sqrt{\frac{3}{2}}}{\pi} + \frac{\pi}{24\sqrt{6}}}{\sqrt{n}} + \frac{\frac{1}{16} + \frac{\pi^2}{6912}}{n} - \frac{\frac{\sqrt{\frac{3}{2}}}{16\pi} + \frac{\pi}{384\sqrt{6}} + \frac{\pi^3}{497664\sqrt{6}}}{n^{3/2}} + \frac{\frac{5}{1536} + \frac{5\pi^2}{497664} + \frac{\pi^4}{286654464}}{n^2} - \frac{\frac{5}{512\sqrt{6}\,\pi} + \frac{5\pi}{36864\sqrt{6}} + \frac{5\pi^3}{31850496\sqrt{6}} + \frac{\pi^5}{34398535680\sqrt{6}}}{n^{5/2}} + \frac{\frac{35}{221184} + \frac{35\pi^2}{63700992} + \frac{7\pi^4}{22932357120} + \frac{\pi^6}{29720334827520}}{n^3}\right)$$

This expansion also follows from an expansion of the BesselI function (see [4]) and the formula

$$A000041(n) \sim \frac{2\,\pi\, BesselI\left(\frac{3}{2}, \sqrt{24\,n-1} * \frac{\pi}{6}\right)}{(24\,n-1)^{3/4}}$$

```
besseliasy[r_, z_] :=
  Exp[z] / Sqrt[2 Pi * z] *
   (1 - (4 r^2 - 1) / (8 z) + (4 r^2 - 1) * (4 r^2 - 9) / (2! (8 z) ^2) -
     (4 r^2 - 1) * (4 r^2 - 9) * (4 r^2 - 25) / (3! (8 z) ^3));

2 * Pi * besseliasy[3 / 2, Sqrt[24 n - 1] * Pi / 6] / (24 n - 1) ^ (3 / 4)
```

$$\frac{2\sqrt{3}\, e^{\frac{1}{6}\sqrt{-1+24\,n}\,\pi}\left(1 - \frac{6}{\sqrt{-1+24\,n}\,\pi}\right)}{-1+24\,n}$$

```
Simplify[Normal[Series[% / (E^(Sqrt[2/3] Sqrt[n] Pi) / (4 Sqrt[3] n)), {n, Infinity, 1}]]]
```

$$1 - \frac{72+\pi^2}{24\sqrt{6}\sqrt{n}\,\pi} + \frac{432+\pi^2}{6912\,n}$$

We have a constants

$$r1 = \pi\sqrt{\frac{2}{3}} \qquad b1 = 1 \qquad c1 = -\frac{\sqrt{\frac{3}{2}}}{\pi} - \frac{\pi}{24\sqrt{6}}$$

The second generating function is

$$\prod_{k=1}^{\infty} \frac{1}{(1-x^{2k})}$$

and the second asymptotics (for even powers) we get from the previous formula after substitution $n \rightarrow \frac{n}{2}$

$$\frac{e^{\sqrt{\frac{2}{3}}\sqrt{n}\,\pi}\left(1-\frac{\frac{\sqrt{\frac{3}{2}}}{\pi}+\frac{\pi}{24\sqrt{6}}}{\sqrt{n}}\right)}{4\sqrt{3}\,n}\ /.\ n\to n/2$$

$$\frac{e^{\frac{\sqrt{n}\,\pi}{\sqrt{3}}}\left(1-\frac{\sqrt{2}\left(\frac{\sqrt{\frac{3}{2}}}{\pi}+\frac{\pi}{24\sqrt{6}}\right)}{\sqrt{n}}\right)}{2\sqrt{3}\,n}$$

$$r2=\frac{\pi}{\sqrt{3}} \qquad b2=1 \qquad c2=-\sqrt{2}\left(\frac{\sqrt{\frac{3}{2}}}{\pi}+\frac{\pi}{24\sqrt{6}}\right)$$

After substitution of r1,b1,c1 and r2,b2,c2 into the formula we get an expression

$$\text{Expand}\left[\text{Simplify}\left[\text{minorsqrt}\left[\sqrt{\frac{2}{3}}\,\pi,\ 1,\ -\frac{\sqrt{\frac{3}{2}}}{\pi}-\frac{\pi}{24\sqrt{6}},\ \frac{\pi}{\sqrt{3}},\ 1,\ -\sqrt{2}\left(\frac{\sqrt{\frac{3}{2}}}{\pi}+\frac{\pi}{24\sqrt{6}}\right)\right]\right]\right]$$

$$-\frac{15}{8\pi}-\frac{\pi}{16}$$

```
N[%, 60]
-0.793180577443969586537229324101897787891546758987655796614687
```

The final asymptotic is

$$\frac{e^{\pi\sqrt{n}}}{8\,n^{5/4}} * \left(1-\frac{\frac{\pi}{16}+\frac{15}{8\,\pi}}{\sqrt{n}}\right)$$

Note that for a sequence A000009 with the generating function

$$\prod_{k=1}^{\infty}(1+x^k)$$

we have a similar expansion

$$\frac{1}{4\times 3^{1/4}\,n^{3/4}}\,e^{\frac{\sqrt{n}\,\pi}{\sqrt{3}}}\left(1+\frac{-\frac{3\sqrt{3}}{8\pi}+\frac{\pi}{48\sqrt{3}}}{\sqrt{n}}+\frac{-\frac{5}{128}-\frac{45}{128\pi^2}+\frac{\pi^2}{13824}}{n}+\frac{-\frac{315\sqrt{3}}{1024\pi^3}+\frac{35\sqrt{3}}{2048\pi}-\frac{35\pi}{36864\sqrt{3}}+\frac{\pi^3}{1990656\sqrt{3}}}{n^{3/2}}+\frac{\frac{105}{65536}-\frac{42525}{32768\pi^4}+\frac{315}{16384\pi^2}-\frac{7\pi^2}{1769472}+\frac{\pi^4}{1146617856}}{n^2}\right)$$

Numerical verification, the asymptotic ratio tends to 1**:**

Show[ListPlot[Table[A002513[[n]] / $\left(\dfrac{e^{\sqrt{n}\,\pi}\left(1-\dfrac{\frac{15}{8\pi}+\frac{\pi}{16}}{\sqrt{n}}\right)}{8\,n^{5/4}}\right)$,
  {n, 1, Length[A002513]}]],
 Plot[1, {n, 1, Length[A002513]}, PlotStyle → Red]]



Richardson extrapolation, from 100000 terms of the sequence. The convergence is very good.

$MaxExtraPrecision = 1000;

funs[n_] := A002513[[n]] / $\left(\dfrac{e^{\sqrt{n}\,\pi}\left(1-\dfrac{\frac{15}{8\pi}+\frac{\pi}{16}}{\sqrt{n}}\right)}{8\,n^{5/4}}\right)$;

```
Do[
 Print[
  N[Sum[(-1)^(m+j)*funs[j*Floor[Length[A002513]/m]]*
     j^(m-1)/(j-1)!/(m-j)!, {j, 1, m}], 40]], {m, 10, 200, 10}]

0.9999999999007739882370509816807619536984
0.9999999999668488887365625833353730059168
0.9999999999822704690609118701133205704664
0.9999999999885860089615093620834664950522
0.9999999999918749766337564384627573823552
0.9999999999938365402446391111515780420585
0.9999999999951208331974891126577583702234
```

References:

[1] OEIS - The On-Line Encyclopedia of Integer Sequences
[2] V. Kotěšovec, A method of finding the asymptotics of q-series based on the convolution of generating functions, arXiv:1509.08708 [math.CO]
[3] G. H. Hardy and S. Ramanujan, Asymptotic formulae in combinatory analysis, Proc. London Math. Soc., 1917, 75–115
[4] Wikipedia, Bessel function

This article was published on the website http://www.kotesovec.cz/math.htm, 24.8.2019
and in the OEIS, Aug 24 2019